


Hromadná změna datumu a času v buňkách


Hromadná změna datumu a času v buňkách je realizovatelná ručním přepisem, funkcemi DATUM a ČAS (ve VBA funkce DateSerial a TimeSerial), případně funkcí EDATE a poslepujím dalších funkcí. Připomínám, že pro aktuální datum slouží klávesová zkratka Ctrl + ; (středník), a pro čas Ctrl + Shift + : (dvojtečka). Na listu daný úkol zvládají i funkce DNES a NYNÍ (čas je rozdíl NYNÍ-DNES), ve VBA pak máme funkce Date a Now, resp. Time.



43562
30.01.2019
21:57
6:59
24.04.2019

Hromadná změna
datumu a času v
buňkách

Ruční úpravy datumu a času ve více buňkách jsou asi nejotravnější možností. Na numerickém bloku klávesnice nemáme k dispozici ani tečku pro datum, ani dvojtečku pro čas (u ní navíc potřebujeme Shift). U datumu si můžeme pomoci tak, že kupříkladu vstup 9/1 do (předem naformátované) buňky Excel zamění za 9.1. aktuálního roku (9.1.2019). Pro čas se hodí přidat nahrazení dvou čárek za dvojtečku do Automatických oprav (viz Soubor / Možnosti / Kontrola pravopisu a mluvnice / tlačítko Možnosti automatických oprav). Zápis 9,,26 bude tedy nahrazen za 9:26.



Automatické nahrazování dvou čárek za dvojtečku

Celé je to ale kostrbaté a pomalé. My se proto dnes naučíme, jak posouvat datum a čas oběma směry s využitím maker a vestavěné funkce `DateAdd`, jež nabízí ve svém prvním parametru řadu možností pro posun jak v kalendáři, tak v čase obecně. Lehce jsem se přitom inspiroval článkem [Entering Times Into Cells](#).

V případě datumu nebudeme ošetřovat případ, kdy by překročilo hranici 1.1.1900, což je na listu nejzazší datum (zatímco ve VBA je jím 1.1.100). U času už si ovšem musíme pohlídat překlápění přes půlnoc. A problémů je u něj daleko více. Ve VBA existuje jen jeden datový typ společný datumu a času – `Date`. Při vracení hodnot typu `Date` zpět do listu můžeme narazit na problémy s formátem buňky. Proto je jistější vracet interní hodnotu – pro datum je jím velké celé číslo a pro čas desetinné číslo. Pro algoritmus se tedy hodí matematické a konverzní funkce. Musíme si přitom uvědomit rozdíly mezi funkcemi `Int` a `CInt`, a stejně tak přibrzdit při myšlence použít operátor `Mod` (který se chová jinak, než funkce `MOD` na listu).

```
1  Sub DatumPosun(Jednotka As String, Posun As Integer)
2
3  Dim varPole() As Variant
4  Dim varPolozka As Variant
5  Dim lngHodnota As Long
6
7  'prvni parametr pro DateAdd
8  'yyyy ... Year
9  'q ... Quarter
10 'm ... Month
11 'y ... Day of year
12 'd ... Day
13 'w ... Weekday
14 'ww ... Week
15
16 'pokud neni vybrana oblast bunek...
17 If TypeName(Selection) <> "Range" Then
18     'opusteni procedury
19     Exit Sub
20 End If
21
22 'vypnuti prepoctu listu
23 Application.Calculation = xlCalculationManual
24
25 's vyberem...
26 With Selection
27
28     'redimenzovani pole
29     ReDim varPole(1 To .Rows.Count, 1 To .Columns.Count)
30
31     'pro kazdou bunku vyberu
32     For i = 1 To UBound(varPole, 1)
33         For j = 1 To UBound(varPole, 2)
34             'test na datum
35             If IsDate(.Cells(i, j)) Then
36                 'prepocet datumu
37                 lngHodnota = CLng(DateAdd(Jednotka, Posun, _
38                     .Cells(i, j)))
39                 varPole(i, j) = lngHodnota
40                 'pokud v bunce neni datum
41                 Else
42                     'pouze prevzeti obsahu bunky
43                     varPole(i, j) = .Cells(i, j)
44                 End If
45             Next j
46         Next i
47
48     End With
49
50     'preklopeni pole zpet do listu
51     Selection = varPole
52
53     'zapnuti prepoctu listu
54     Application.Calculation = xlCalculationAutomatic
55
56 End Sub
```

```

1  Sub CasPosun(Jednotka As String, Posun As Integer)
2
3  Dim varPole() As Variant
4  Dim varPolozka As Variant
5  Dim dblHodnota As Double
6
7  'prvni parametr pro DateAdd
8  'h ... Hour
9  'n ... Minute
10 's ... Second
11
12 'pokud neni vybrana oblast bunek...
13 If TypeName(Selection) <> "Range" Then
14     'opusteni procedury
15     Exit Sub
16 End If
17
18 'vypnuti prepoctu listu
19 Application.Calculation = xlCalculationManual
20
21 's vyberem...
22 With Selection
23
24     'redimenzovani pole
25     ReDim varPole(1 To .Rows.Count, 1 To .Columns.Count)
26
27     'pro kazdou bunku vyberu
28     For i = 1 To UBound(varPole, 1)
29         For j = 1 To UBound(varPole, 2)
30             'test na cas ... Dick Kusleika
31             If IsDate(.Cells(i, j).Text) And Not IsDate(.Cells(i, _
32                 j).Value) Then
33                 'prepocet casu
34                 dblHodnota = Abs(CDbl(DateAdd(Jednotka, Posun, _
35                     .Cells(i, j))))
36                 varPole(i, j) = dblHodnota - Int(dblHodnota)
37             Else
38                 'pouze prevzeti obsahu bunky
39                 varPole(i, j) = .Cells(i, j)
40             End If
41         Next j
42     Next i
43
44 End With
45
46 'preklopeni pole zpet do listu
47 Selection = varPole
48
49 'zapnuti prepoctu listu
50 Application.Calculation = xlCalculationAutomatic
51
52 End Sub

```

Pozn. Jestliže na listu chybí ekvivalent VBA funkce DateAdd, pak list vrací úder v podobě funkce WORKDAY, která pro změnu nemá sestřičku pod VBA. Musíme si ji tedy buď doprogramovat, nebo si ji vypůjčit přes WorksheetFunction.WorkDay (což je lepší varianta, protože případné svátky je také výhodnější generovat na listu, v reálu řekněme plus minus jeden rok vůči aktuálnímu roku). Níže uvedený kód čerpá svátky z pojmenované oblasti buněk – definovaný název Svatky (list Výpis svátků z přílohy).

```
1 Sub PracovniDenPosun(Posun As Integer)
2
3 Dim varPole() As Variant
4 Dim varPolozka As Variant
5 Dim lngHodnota As Long
6
7 'pokud není vybrána oblast buněk...
8 If TypeName(Selection) <> "Range" Then
9     'opuštění procedury
10    Exit Sub
11 End If
12
13 'vypnutí přepočtu listu
14 Application.Calculation = xlCalculationManual
15
16 's vyberem...
17 With Selection
18
19     'redimenzování pole
20    ReDim varPole(1 To .Rows.Count, 1 To .Columns.Count)
21
22    'pro každou buňku vyberu
23    For i = 1 To UBound(varPole, 1)
24        For j = 1 To UBound(varPole, 2)
25            'test na datum
26            If IsDate(.Cells(i, j)) Then
27                'přepočítání datumu
28                lngHodnota = _
29                    CLng(WorksheetFunction.WorkDay(.Cells(i, j), _
30                        Posun, Range("Svatky")))
31                varPole(i, j) = lngHodnota
32            'pokud v buňce není datum
33            Else
34                'pouze převzetí obsahu buňky
35                varPole(i, j) = .Cells(i, j)
36            End If
37        Next j
38    Next i
39
40 End With
41
42 'překlopení pole zpět do listu
43 Selection = varPole
44
45 'zapnutí přepočtu listu
46 Application.Calculation = xlCalculationAutomatic
47
48 End Sub
```

Nyní již vám nic nebrání zpracovat vlastní varianty procedur a řídit tak posuny pro datum a čas programově.

1	Sub TestDatumCasPosun()
2	
3	Call DatumPosun("ww", 1)
4	'Call CasPosun("h", -1)
5	'Call PracovniDenPosun(-3)
6	
7	End Sub

Ke stažení

[datum_cas_posun.zip](#)